

Equipment

The Background

The purpose of this project is to calculate the standing wave functions of a quantum mechanical particle trapped in a finite potential well. Technically, we will be solving the time-independent Schrödinger equation:

$$-\frac{\hbar^2}{2m} \frac{d^2}{dx^2} \psi = (E - V) \psi \quad (1)$$

We will walk through what you need to know for this project in this first section.

First case: E greater than V

There are two important cases to consider when discussing Equation 1. The first is when $E > V$. This means that the total energy of the particle is greater than its potential energy. The remainder is its kinetic energy. Although we could replace $E - V$ with a symbol for kinetic energy, it is typical not to. You will see why when we discuss the second case below.

The general solution of Schrödinger's equation 1 in this first case is

$$\psi(x) = A \cos(kx + \phi) \quad (2)$$

The value of k is directly related to the energy of the particle:

$$\hbar k = \sqrt{2m(E - V)} \quad (3)$$

Exercise

Calculate the k value of an electron with an energy of one electron-volt. Assume the electron is free from any force potential ($V = 0$).

This k -value is actually the reciprocal of the de Broglie wavelength of the particle. It is sometimes called the “propagation vector”.

Exercise

Derive Equation 3 from the de Broglie wavelength formula $\lambda = h/p$. You will need to use the (non-relativistic) definitions of momentum and kinetic energy and the relationship $k = 1/\lambda$.

What about A and ϕ ? These are determined by the nature of the potential and any boundary conditions we have in our particular problem. The bulk of this project will be in determining their proper value.

Second case: E less than V

The second case to deal with in solving the Schrödinger equation is when $E < V$. This is relevant because we are discussing a particle “trapped” in a potential well. There must be some x -values where total energy is under the potential.

But this implies that the kinetic energy is negative. In a classical world, the particle is never in this region since $KE = \frac{1}{2}mv^2$ which cannot be smaller than

zero. But in the quantum world there is a non-zero solution to Equation 1 given by

$$\psi(x) = A \exp(\kappa x + \phi) \quad \text{or} \quad \psi(x) = A \exp(-\kappa x + \phi) \quad (4)$$

where

$$\hbar\kappa = \sqrt{2m(V - E)} \quad (5)$$

Like before, the A and ϕ are determined by boundary conditions. Notice that the solution is either one of these equations—never both.¹ The choice is usually obvious since we want to pick the one that produces exponential decay as the x gets large rather than exponential growth.

Summary

That is all you will need to know about solving the Schrödinger equation for this project. The basic rule is that the solution is a simple cosine wave when the kinetic energy is positive. The wavelength is related to this kinetic energy via the de Broglie wavelength in Equation 3. In the region where the potential energy exceeds the total energy of the particle (the “forbidden region”), the wave exponentially decays away.

Setting up the Spreadsheet

The potential energy function V in which we are interested is defined by the blue line in Figure 1.




Figure 1: The potential well for our particle

Since we are going to discuss a particle that is “trapped” in this potential, we should consider the energy of the particle to be less than two. For the sake of argument, let’s assume the energy value is one. (We will change this later.) This is represented in Figure 1 by the red line.

Getting Started

Let’s start building our spreadsheet to represent this potential. I’ve chosen to use OpenOffice.org Calc for this project. This open source software works very similar to Microsoft Excel. There are many ways to get the program if you choose to use it. Go to <http://www.openoffice.org> for more information.² Although some of the details will be different, you can use Excel if you prefer.

Since the screen real estate on my little netbook is limited, I have eliminated all the toolbars and will be using the menus in this walk-through. Figure 2 shows the About page from the Help menu.

¹I’m not 100% sure of this statement, but I can’t recall a solution in my studies that used both at the same time. It will certainly be true in *this* project.

²You can even get a completely portable version to install and carry around on your flash drive. See http://portableapps.com/apps/office/openoffice_portable.




Figure 2: OpenOffice.org Calc

To begin, type the data from Table 1 into the appropriate cells.

A1	\hbar
A2	m
A3	E
B1	1
B2	1
B3	1

Table 1: The control panel

We will use the first few rows as a kind of “control panel” for our document. Notice that we have set \hbar (symbol: \hbar) to be equal to one. This is a common practice among physicists and means that we are no longer working in SI units. We could put the standard value in here, but the numbers get a bit unruly (a mere annoyance, but annoying nonetheless). We could also put the mass of the electron in for m , but let’s keep it simple. This implies that the E we are using is some non-standard unit as well. But since we are really interested in the relative differences, this will be good enough.

Okay—enough excuses. Let’s go ahead and build our potential function. The first step is to input the data from Table 2.

A8	x_1
B8	V_1
C8	k_1
D8	wf_1
A9	-10
B9	2
C9	$=\text{SQRT}(2*\$B\$2*(B9-\$B\$3))/\$B\1
D9	$=\text{EXP}(C9*A9)$

Table 2: Potential function: left side

When you are done, your screen should look something like Figure 3.




Figure 3: Initial setup

The first column is our horizontal distance on the graph and the second column is the potential energy at that point. The reason for the “1” suffix is that we will split this problem into three parts:

- The left side with $V = 2$,
- The center with $V = 0$, and
- The right side with $V = 2$ again.

The third column represents our k value and corresponds to equation 5. The fourth column holds our wavefunction—see equation 4.

Exercise

Explain why we choose the $\exp(\kappa x)$ form for the left side of the wave function. Be sure to discuss the choice of the sign in front of κ .

Now we need to push these calculations to $x = -5$. Enter the following from Table 3.

A10	=A9+0.5
B10	=B9
C10	=SQRT(2*\$B\$2*(B10-\$B\$3))/B\$1
D10	=EXP(C10*A10)

Table 3: Building up the left side

Now drag these cells down to row 19. Do this by selecting the range A10:D19 and type **Ctrl-D**. Your screen should now look like Figure 4.




Figure 4: Left side built

You may not be showing as many decimals as I am. One simple way to match my picture is to click on the small box to the left of column A and above 1 which will highlight all the cells in the sheet. Now go to the menu item **Format >> Cells....** Click on the **Numbers** tab in the pop-up window. Click **Number** in the **Category** selection box and **-1234.1235** in the **Format** selection box. The **Format** code on the bottom should read **0.0000** (see Figure 5). Click **OK** and you should be set.




Figure 5: Changing the number format

Now might be a good time to save the spreadsheet (if you haven't done so already). Notice that OpenOffice Calc will automatically save the document in Open Document Format (extension **.ods**). Microsoft Excel might refuse to read this format, so I suggest you would save the document in an Excel format by changing the file type to **.xls** when you type in your filename.

We now need to build the center and right sides of our potential well. Enter the data from Table 4.

A22	x2
B22	V2
C22	k2
D22	wf2
A23	-5
B23	0
C23	=SQRT(2*\$B\$2*(\$B\$3-B23))/ \$B\$1
D23	=COS(C23*A23)
A24	=A23+0.5
B24	=B23
C24	=SQRT(2*\$B\$2*(\$B\$3-B24))/ \$B\$1
D24	=COS(C24*A24)

Table 4: Building up the center

Note the differences in the formulas for columns C and D. Now drag the cell in row 24 down to row 43 by highlighting cells A24:D43 and typing **Ctrl-D**. Your screen should look like Figure 6.

	A	B	C	D	E	F	G	H	I	J	K
21											
22	x2	V2	k2	wf2							
23	-5.0000	0.0000	1.4142	0.7053							
24	-4.5000	0.0000	1.4142	0.9967							
25	-4.0000	0.0000	1.4142	0.8102							
26	-3.5000	0.0000	1.4142	0.2351							
27	-3.0000	0.0000	1.4142	-0.4527							
28	-2.5000	0.0000	1.4142	-0.9234							
29	-2.0000	0.0000	1.4142	-0.9514							
30	-1.5000	0.0000	1.4142	-0.5231							
31	-1.0000	0.0000	1.4142	0.1559							
32	-0.5000	0.0000	1.4142	0.7602							
33	0.0000	0.0000	1.4142	1.0000							
34	0.5000	0.0000	1.4142	0.7602							
35	1.0000	0.0000	1.4142	0.1559							
36	1.5000	0.0000	1.4142	-0.5231							
37	2.0000	0.0000	1.4142	-0.9514							
38	2.5000	0.0000	1.4142	-0.9234							
39	3.0000	0.0000	1.4142	-0.4527							
40	3.5000	0.0000	1.4142	0.2351							
41	4.0000	0.0000	1.4142	0.8102							
42	4.5000	0.0000	1.4142	0.9967							
43	5.0000	0.0000	1.4142	0.7053							
44											
45											

Figure 6: Center built

Now the same story for the right side. Enter the data from Table 5 and drag these cells down to row 57. Your screen should look like Figure 7.

A46	x2
B46	V2
C46	k2
D46	wf2
A47	-5
B47	0
C47	=SQRT(2*\$B\$2*(B47-\$B\$3))/ \$B\$1
D47	=EXP(-C47*A47)
A48	=A47+0.5
B48	=B47
C48	=SQRT(2*\$B\$2*(B48-\$B\$3))/ \$B\$1
D48	=EXP(-C48*A48)

Table 5: Building up the right side

	A	B	C	D	E	F	G	H	I	J	K
34	0.5000	0.0000	1.4142	0.7602							
35	1.0000	0.0000	1.4142	0.1559							
36	1.5000	0.0000	1.4142	-0.5231							
37	2.0000	0.0000	1.4142	-0.9514							
38	2.5000	0.0000	1.4142	-0.9234							
39	3.0000	0.0000	1.4142	-0.4527							
40	3.5000	0.0000	1.4142	0.2351							
41	4.0000	0.0000	1.4142	0.8102							
42	4.5000	0.0000	1.4142	0.9967							
43	5.0000	0.0000	1.4142	0.7053							
44											
45											
46	x3	V3	k3	wf3							
47	5.0000	2.0000	1.4142	0.0008							
48	5.5000	2.0000	1.4142	0.0004							
49	6.0000	2.0000	1.4142	0.0002							
50	6.5000	2.0000	1.4142	0.0001							
51	7.0000	2.0000	1.4142	0.0001							
52	7.5000	2.0000	1.4142	0.0000							
53	8.0000	2.0000	1.4142	0.0000							
54	8.5000	2.0000	1.4142	0.0000							
55	9.0000	2.0000	1.4142	0.0000							
56	9.5000	2.0000	1.4142	0.0000							
57	10.0000	2.0000	1.4142	0.0000							
58											

Figure 7: Left side built

Tailoring the Spreadsheet

We now have the backbone of the spreadsheet built and we are about half done with the setup. However, there is a problem in what we have done so far. The wavefunction in column D is not continuous—it's a broken string. Notice how the functions do not touch at the interfaces when $x = -5$ or $x = 5$. We need to “tie” these three functions together in such a way that the whole curve is smooth and connected.

Type the information³ from Table 6 and drag cell E9 down to row 19, cell E23 to row 43, and cell E47 to row 57.

³Psi is the greek letter ψ which is traditionally used to symbolize the wave function.

E7	1
E8	psi1
E9	=\$E\$7*D9
E21	1
E22	psi2
E23	=\$E\$21*D23
E45	1
E46	psi3
E47	=\$E\$45*D47

Table 6: Copying the wave function

It doesn't look like we have done anything—we've simply copied our wave function into column E. But this is how we are going to “stitch” the functions together. Type the information from Table 7. After that, your screen should look like Figure 8.

E7	=D23/D19
E45	=D43/D47

Table 7: Stitching up the functions




Figure 8: Functions are now “stitched” together

It is important to note that the numbers in cells E19 and E23 are the same. The same should be true of cells E43 and E47. It is now time to build a chart to see what we are working with.

Charting The Wave Function

After saving your work (you have been saving regularly, haven't you?), move your cursor up to cell N1. Now go to the menu and click **Insert >> Chart...**

A new window should pop up. In step 1, choose the chart type labeled **XY (Scatter)**, click on the **Lines Only** picture, and the check box **Smooth lines**. See Figure 9.




Figure 9: Choosing a chart

Click **Next >>**. Skip step 2 by clicking on **Next >>** again. Now click the **Add** button. You should see something like Figure 10.




Figure 10: Adding data series

Click on **Name** in the **Data ranges** selection box and type **psi1** in the **Range for Name** input box. Then click on **X-Values** in the **Data ranges** selection box and click on the button on the far right next to the **Range for X-Values**. The pop-up box disappears temporarily and will now allow you to select a range on your spreadsheet. Highlight the cells **A9:A19**. The pop-up box will reappear and now reads **\$Sheet1.\$A\$9:\$A\$19** in the **Range for X-Values** box. You can also simply type this in directly. Now click on the **Y-Values** in the **Data ranges** selection box. Highlight cells **E9:E19**. The left side of the wave function is now linked to the chart.

Repeat this procedure for the center of the wave functions. Do this by clicking **Add** under the **Data series** selection box. A new **Unnamed Series** should appear. Type **psi2** as the name and link the **X-Values** to cells **A23:A43** and the **Y-Values** to cells **E23:E43**.

Do this again for the right side. Name it **psi3** and link it to cells **A47:A57** and **E47:E57**. After all this, click **Finish** and scroll to the top to see your chart. Click anywhere off of the chart to get out of "charting mode." You might want to zoom out a bit to see it all on one screen. (Do this under menu item **View >> Zoom...**) Hopefully it looks something like Figure 11.




Figure 11: Spreadsheet with chart

These defaults are nice, but we are going to change some to show off our curve a bit more. Click on the chart (this puts you in something like an “object mode”). Then double-click on the chart (this puts you in something like a “chart mode”). Right click and a small contextual menu pops up. You could click on **Data Ranges...** to correct any errors you may have made while linking the chart.

But now I want to make three adjustments. First: get rid of the legend. Right click on it and click **Cut**. Or left-click and hit the **Del** key.

Second thing is to do is to change the x -axis. Double-click on one of the numbers. A pop-up box will appear. Go to the **Scale** tab and de-select the **Minimum**, **Maximum**, and **Major Interval** check boxes labeled **Automatic**. Change the numbers to -10, 10, and 5, respectively. Now go to the **Numbers** tab. De-select the **Source Format** check box and change the **Format Code** to just a zero (0). This will suppress all of the unnecessary decimals in the chart. Click **OK**.

Finally double-click on one of the y -axis numbers and change its **Format Code** (on the **Numbers** tab) to zero also. Click anywhere off the chart to get out of charting mode. Your chart is now ready. Check it against Figure 12.




Figure 12: Spreadsheet with adjusted chart

Now not only do we have a nice looking chart of our wave function, but we also have the classically “forbidden” areas highlighted in a different color.

Another complication

You might think we are now done. We have nice looking wave function for our particle with energy $E = 1$ trapped in a potential $V = 2$ units high. And we do. But observe what happens when we change the energy value to, say, $E = 0.5$. Do this by changing cell B3 to 0.5. Your spreadsheet will look like Figure 13.




Figure 13: Changing the energy to 0.5 units

The problem is that the wave cannot have a spike like it does at $x = -5$ and $x = 5$. The function must be both continuous and smooth. At the interface the functions must not only have the same value, but they must also have the same slope.

To explain why, imagine this were a physical string. The spike in the curve is indicating that this energy level creates a reflected wave that is destructively out-of-phase with the original. The reflected wave destroys rather than resonates into a standing wave. In other words, this particular energy level cannot support a standing wave. Remember de Broglie's explanation of the Bohr atom? Only certain energy levels can produce standing waves. The same is true here.

Does that mean that our spreadsheet is broken? On the contrary, it is doing precisely what it should. It is telling us that this is an invalid energy level for our trapped particle. We must look for another energy level. We will know when we find one when the slopes at the boundaries match. Try $E = 0.4$. This is much closer to a valid energy level. Is there a way to measure how close we are? I'm glad you asked. Type the information from Table 8 into the spreadsheet and drag the formulas down to fill in the rest of the table. Your document should look something like Figure 14.

D1	nudge
E1	1e-9
F8	npsi1
F9	=\$E\$7*EXP(C9*(A9+\$E\$1))
G8	m1
G9	=(F9-E9)/\$E\$1
F22	npsi2
F23	=\$E\$21*COS(C23*(A23+\$E\$1))
G22	m2
G23	=(F23-E23)/\$E\$1
F46	npsi3
F47	=\$E\$45*EXP(-C47*(A47+\$E\$1))
G46	m3
G47	=(F47-E47)/\$E\$1

Table 8: Preparing to measure the slopes

	A	B	C	D	E	F	G	H	I	J	K	L	M	N
31	-1.0000	0.0000	0.8944	0.6260	0.6260	0.6260	0.6975							
32	-0.5000	0.0000	0.8944	0.9017	0.9017	0.9017	0.3868							
33	0.0000	0.0000	0.8944	1.0000	1.0000	1.0000	0.0000							
34	0.5000	0.0000	0.8944	0.9017	0.9017	0.9017	-0.3868							
35	1.0000	0.0000	0.8944	0.6260	0.6260	0.6260	-0.6975							
36	1.5000	0.0000	0.8944	0.2272	0.2272	0.2272	-0.8710							
37	2.0000	0.0000	0.8944	-0.2163	-0.2163	-0.2163	-0.8732							
38	2.5000	0.0000	0.8944	-0.6173	-0.6173	-0.6173	-0.7037							
39	3.0000	0.0000	0.8944	-0.8968	-0.8968	-0.8968	-0.3957							
40	3.5000	0.0000	0.8944	-0.9999	-0.9999	-0.9999	-0.0099							
41	4.0000	0.0000	0.8944	-0.9064	-0.9064	-0.9064	0.3778							
42	4.5000	0.0000	0.8944	-0.6346	-0.6346	-0.6346	0.6913							
43	5.0000	0.0000	0.8944	-0.2379	-0.2379	-0.2379	0.8687							
44														
45														
46	x3	v3	k3	vf3	psi3	npsi3	m3							
47	5.0000	2.0000	1.7889	0.0001	-0.2379	-0.2379	0.4269							
48	5.5000	2.0000	1.7889	0.0001	-0.0973	-0.0973	0.1740							
49	6.0000	2.0000	1.7889	0.0000	-0.0398	-0.0398	0.0711							
50	6.5000	2.0000	1.7889	0.0000	-0.0163	-0.0163	0.0291							
51	7.0000	2.0000	1.7889	0.0000	-0.0066	-0.0066	0.0119							
52	7.5000	2.0000	1.7889	0.0000	-0.0027	-0.0027	0.0049							
53	8.0000	2.0000	1.7889	0.0000	-0.0011	-0.0011	0.0020							
54	8.5000	2.0000	1.7889	0.0000	-0.0005	-0.0005	0.0008							
55	9.0000	2.0000	1.7889	0.0000	-0.0002	-0.0002	0.0003							
56	9.5000	2.0000	1.7889	0.0000	-0.0001	-0.0001	0.0001							
57	10.0000	2.0000	1.7889	0.0000	0.0000	0.0000	0.0001							
58														
59														
60														
61														

Figure 14: Adding the slopes to the spreadsheet

What have we done here? Remember that we want to be able to compare the slopes of the three curves at their interfaces. This is what is in column G. It is the basic rise-over-run formula. But instead of measuring the rise and run from point to point in the rows we have already calculated, we are measuring the rise over a *very small* run. That is what the “nudge” value is in cell E1 (notice how it is the denominator in the slope calculation). The values in column F are the wavefunction evaluated at x plus this small nudge factor (hence “npsi” or the nudged psi value). The difference between columns F and E is the rise of our slope.

The bottom line here is that we want cells G19 and G23 to match just like we wanted cells E19 and E23 to match. Similarly we want G43 and G47 to match. Unfortunately there is no simple algebraic way to stitch these together like we did in column E. We just have to hunt for a solution.

But just because we are hunting doesn’t mean we can’t be efficient. Type the following from Table 9

G1	dm12
H1	=ABS(G19-G23)
G2	dm23
H2	=ABS(G43-G47)
G3	sbz
H3	=SUM(H1:H2)

Table 9: Checking the slope match

What we have here is a quick table indicating how far off our slopes are from matching. Cell H1 is comparing the slopes at $x = -5$ and cell H2 is comparing the slopes at $x = 5$. Both of these should be zero. Cell H3 simply sums these two in order to get a single measure for us to track. The label **sbz** stands for “should be zero.” Your spreadsheet should look like Figure 15. This spreadsheet is now pretty much ready to use.




Figure 15: Slope double-checks are in place

In order to find our standing wave patterns all we need to do is tweak the energy level in cell B3 until cell H3 goes to zero. For example, $E = 0.36$ is the best fit with two decimals. $E = 0.365$ is the best fit with three decimals.

Exercise

What is the best fit near $E = 0.365$ to four decimals?

Exercise

There is another standing wave pattern around $E = 0.03$. What is its energy out to four decimals?

Find the Energy Levels

Efficient Hunting: Goal Seek

I have one more trick up my sleeve. Change the energy level to 0.01 units. Now go to the menu and click on Tools >> Goal Seek.... Type in H3 for the Formula cell, type 0.001 for the Target value, and type B3 for the Variable cell. The pop-up window should look like Figure 16.




Figure 16: Goal seeking

Click OK. A new pop-up window should read **Goal Seek successful**. Insert result (0.0408) into current cell? Click Yes. The program just found your first energy level. The goal seek is designed to drive the formula cell to the value you specify by changing the variable cell. This is exactly what we need to find our energy levels. Note that you must be fairly close to the goal in order for this to work. But if you can get close, the program will do the rest.

Let's try $E = 1$ again. Type 1 into cell B3 and goal seek it. You will find that the actual answer is not exactly one.

Exercise

Find all the energy levels to four digit accuracy. There are four total. Print out the standing wave pattern for each of them. Go ahead and record these energy levels somewhere (I've left room in row 5) so you don't have to hunt for them again.}




Figure 17: Cosine wave patterns

The Other Half

This is pretty slick, but one problem remains. We are only finding half of the possible energy levels. The patterns you should have so far are in Figure 17.

You won't have the dashed lines, but they are there to remind you that this standing wave pattern oscillates—just like a string. So the antinodes oscillate between the maximum peaks and minimum troughs. Look at the four wave patterns we have found. The number of antinodes is always odd. There are more standing waves with an even number of antinodes that we have yet to find.

Go ahead and save the workbook. Then go to the menu **File >> Save As**. Save a copy of your document with a slightly different name. We are going to make one last adjustment.

When you are ready to work in your copied workbook, make the changes listed in Table 10 and drag them down to row 43.

D23	=SIN(C23*A23)
F23	=\$E\$21*SIN(C23*(A23+\$E\$1))

Table 10: Changing the central cosine to sine

Exercise

Equation 2 is the solution for the wave function in the central region. This involves a cosine function. How can one justify using the sine function that I am proposing in Table 10?

Exercise

Change the energy value to $E = 0.1$ and start goal seeking again. Find these new energy levels to four digit accuracy. There are three total. Print out the standing wave pattern for each of them. Go ahead and type them somewhere like you did for the cosine energy levels. These wave patterns ought to look like those in Figure 18.




Figure 18: Sine wave patterns

Final Wrap-Up

Congratulations! You have successfully solved a fundamental quantum mechanics problem. Two more exercises are left.

Exercise

Figure 19 is a diagram representing the energy levels in our quantum box. This is the same idea as the energy level diagrams we use in explaining the Bohr atom. Redraw this diagram but use the four digit energy levels you have discovered through the course of this project.




Figure 19: Energy level diagram for this quantum box

Exercise

We could easily set up the spreadsheet to analyze a phenomena called “quantum tunneling”. In this case the left and right potentials would be zero and the central potential would be two. Classically, a particle approaching the potential from the left would simply bounce and never get to the right side. But in quantum mechanics, a portion of the wave function does make it through the barrier to the other side. Using Equations 2 and 4 explain how this happens.